
Summary

In part 1 a and b, VM was calculated for both cases by shorting the outputs to any inputs that were changing.  VM was found analytically to be 0.826 V when one input is held high, and to be 1.02 when both inputs switch.  When simulated, VM was 0.875 V for the first case and 1.04 V for the second.  The hand analysis to calculate the average current was found to be 40 μA.  The simulated value was 42.5 μA.
For part 2, the XOR was realized using complimentary pass transistor logic.  After simulating the circuit, it was found that switching B had the highest delay and was therefore the input that caused the worst case delay.
In part 3, the SS process had the longest setup time.  All processes had the same hold time which was in fact zero.  The minimum and maximum clock to q delay increased by a factor of two when a 20fF capacitive load was added to the output.
Part 1a: 

In order to determine VM, when one input is held high and the other switches, we tie node A to VDD and short the output of the NAND gate to input B.  We then solve for VM.

Figure 1
Hand Analysis:

Spice Simulation

Net List File:
* Vm for NAND Gate Simulation – One input switching, the other held high.
vs Vdd GND dc 2

vin a GND pulse(0 2 0 50p 50p 2.5n 5n)

m0 Vout a Vdd Vdd pmos w=1u l=0.25u

m1 Vout Vdd Vdd Vdd pmos w=1u l=0.25u

m2 Vout Vdd c GND nmos w=1u l=0.25u

m3 c a GND GND nmos w=1u l=0.25u

.options post

.option absvar=.025 dvdt relvar=0.05 newtol

.lib '/homes/j/jairam/ECE334/Project/cmos25.models' TT

.op

.tran 0.25n 250n

.end

The results from the simulation show that VM = 0.94 V.

Part 1b:

For 1b, the inputs to the NAND gate switch at the same time.  Therefore, the widths of the 

PMOS transistors can be added since they are in parallel and the lengths of the NMOS transistors can be added since they are in series.  The NAND gate can then be approximated by an inverter.  We therefore have the following circuit.

Figure 3

Spice Simulation:

Net List File:

* Vm for NAND Gate Simulation - Both Inputs Switching

vs Vdd GND dc 2

m0 a a Vdd Vdd pmos w=1u l=0.25u

m1 a a Vdd Vdd pmos w=1u l=0.25u

m2 a a c GND nmos w=1u l=0.25u

m3 c a GND GND nmos w=1u l=0.25u

.options post

.option absvar=.025 dvdt relvar=0.05 newtol

.lib '/homes/j/jairam/ECE334/Project/cmos25.models' TT

.op

.tran 0.25n 250n

.end
A similar graph to figure 2 is generated using spice.  The resulting graph shows that 

VM = 1.04 V.
Part 1c:

Hand Analysis:

Spice Simulation:

Net List File:

* Average Current Drawn From NAND Gate

vs Vdd GND dc 2

vin a GND pulse(0 2 0 50p 50p 2.5n 5n)

m0 Vout a Vdd Vdd pmos w=1u l=0.25u

m1 Vout Vdd Vdd Vdd pmos w=1u l=0.25u

m2 Vout Vdd c GND nmos w=1u l=0.25u

m3 c a GND GND nmos w=1u l=0.25u

Cout Vout GND 100f

.options post

.option absvar=.025 dvdt relvar=0.05 newtol

.lib '/homes/j/jairam/ECE334/Project/cmos25.models' TT

.op

.tran 0.25n 250n

.end
Figure 4 is a plot of the integral of the current drawn from the voltage source, Vs.

Therefore, the average current will be given by the slope of the line.  

Using the two data points on the graph, it can be concluded that the average current is:

This value very closely agrees with the hand analysis.


Part 2:

Logic Function:

Y = A ( B ( C ( D
This logic function can be broken down into a series of cascaded XOR gates.  

i.e. Y = (A ( B) ( (C ( D)

This new logic function is described by the following pass-transistor logic circuit.

[image: image6.png]



Figure 5
Net List File:

* NetList for 4 Input XOR Gate

vs Vdd GND dc 2

va a GND dc 0
vb b GND pulse(0 2 0 1n 1n 50n 100n)
vc c GND dc 0

vd d GND dc 0

* Inverters

map ac a Vdd Vdd pmos l=0.25u w=1u

man ac a GND GND nmos l=0.25u w=1u

mbp bc b Vdd Vdd pmos l=0.25u w=1u

mbn bc b GND GND nmos l=0.25u w=1u

mcp cc c Vdd Vdd pmos l=0.25u w=1u

mcn cc c GND GND nmos l=0.25u w=1u

mdp dc d Vdd Vdd pmos l=0.25u w=1u

mdn dc d GND GND nmos l=0.25u w=1u

m1p 1c 1 Vdd Vdd pmos l=0.25u w=1u

m1n 1c 1 GND GND nmos l=0.25u w=1u

m2p 2c 2 Vdd Vdd pmos l=0.25u w=1u

m2n 2c 2 GND GND nmos l=0.25u w=1u

* Transmission Gates

mt0p 2 b a Vdd pmos l=0.25u w=1u

mt0n 2 bc a GND nmos l=0.25u w=1u

mt1p 2 bc ac Vdd pmos l=0.25u w=1u

mt1n 2 b ac GND nmos l=0.25u w=1u

mt2p y 1 2 Vdd pmos l=0.25u w=1u

mt2n y 1c 2 GND nmos l=0.25u w=1u

mt3p y 1c 2c Vdd pmos l=0.25u w=1u

mt3n y 1 2c GND nmos l=0.25u w=1u

mt4p 1 d c Vdd pmos l=0.25u w=1u

mt4n 1 dc c GND nmos l=0.25u w=1u

mt5p 1 dc cc Vdd pmos l=0.25u w=1u

mt5n 1 d cc GND nmos l=0.25u w=1u

cl y GND 20f

.options post

.option absvar=.025 dvdt relvar=0.05 newtol

.lib '/homes/j/jairam/ECE334/Project/cmos25.models' FS
.op

.tran 0.25n 250n

.end
This sample NetList file was used to simulate the circuit described in figures 6 and 7.  The input voltages (va, vb, vc, and vd) were independently assigned to a pulse signal and the output was simulated in spice for each case.  The resulting signals were plotted and the rise and fall times for each case were measured.  The following graphs illustrate the worst case rise and fall times which is for when the input B changes.

Figure 6 – Worst Case Propagation Delay for B Rising

Figure 7 – Worst Case Propagation Delay for B Falling

The propagation delay for the other inputs, A, C and D had faster responses than the input B.  Therefore the input B represents the worst case propagation delay for the 4-input XOR gate.  

The following table indicates the propagation delay due to each of the four inputs.
	Input
	Rise Time
	Fall Time

	A
	250 ps
	300 ps

	B
	579 ps
	500 ps

	C
	425 ps
	400 ps

	D
	560 ps
	500 ps


Figure 8 – Propagation Delay for FS process
Note: The graphs were only simulated for the ‘FS’ transistor type as indicated in the NetList file.  The lack of licenses for Awaves prevented us from performing a complete analysis over all process corners.
Part 3:

The circuit in figure 9 is that of a positive edge triggered register.  The object of this part is to measure the worst case setup and hold times.

Figure 9 – Positive Edge Triggered Register

NetList File:
* Positive Edge Triggered Register

vs vdd gnd dc 2

vclk ck gnd pulse(0 2 0 50p 50p 2.5n 5n)

vin d gnd pulse(0 2 4820p 50p 50p 5n 10n)

* First Latch

m0 1 d vdd vdd pmos w=2u l=0.25u

m1 qbar ck 1 vdd pmos w=2u l=0.25u

m2 qbar ckbar 3 gnd nmos w=1u l=0.25u

m3 3 d gnd gnd nmos w=1u l=0.25u

m4 q qbar vdd vdd pmos w=1u l=0.25u

m5 q qbar gnd gnd nmos w=0.5u l=0.25u

m6 4 q vdd vdd pmos w=2u l=0.25u

m7 qbar ckbar 4 vdd pmos w=2u l=0.25u

m8 qbar ck 5 gnd nmos w=1u l=0.25u

m9 5 q gnd gnd nmos w=1u l=0.25u

* ck and ckbar

m10 ckbar ck vdd vdd pmos w=2u l=0.25u

m11 ckbar ck gnd gnd nmos w=1u l=0.25u

* Second Latch

m12 10 q vdd vdd pmos w=2u l=0.25u

m13 q2bar ckbar 10 vdd pmos w=2u l=0.25u

m14 q2bar ck 11 gnd nmos w=1u l=0.25u

m15 11 q gnd gnd nmos w=1u l=0.25u

m16 q2 q2bar vdd vdd pmos w=1u l=0.25u

m17 q2 q2bar gnd gnd nmos w=0.5u l=0.25u

m18 12 q2 vdd vdd pmos w=2u l=0.25u

m19 q2bar ck 12 vdd pmos w=2u l=0.25u

m20 q2bar ckbar 13 gnd nmos w=1u l=0.25u

m21 13 q2 gnd gnd nmos w=1u l=0.25u

* Capacitive Load

* cout q2 gnd 20f

.options post

.option absvar=.025 dvdt relvar=0.05 newtol

.lib '/homes/j/jairam/ECE334/Project/cmos25.models' SS
.op

.tran 0.25n 250n

.end
The preceding NetFile simulates the positive edge triggered register for the SS process.  By varying the input ‘d’, the setup time and the hold time for the different processes were tested. 
Setup Time:

 To determine the set up time, the delay to the input was adjusted such that it came very close to the rising edge of the clock.  The delay was then increased incrementally until the register failed to latch the value.  The graphs in figures 10 and 11 illustrate the two timings that succeeded and failed in q latching the value.

Figure 10 – Setup Time Success for SS Process

Figure 11 – Setup Time Failure for SS Process

From the preceding two figures, it can be seen that when the input delay was set to 4.76 ns, the output latched the value and q went high, however, when the input delay was set to 4.77 ns, the output did not latch the value and thus q did not change.  Since the clock rises at 5 ns, this suggests that the setup time is:

5 ns – 4.76 ns = 240 ps

This process was repeated for all 5 processes.  The following table, figure 12, summarizes these results.  From figure 12, we can easily see that the SS process has the worst case setup time.
	Process
	Success
	Failure
	Setup Time

	TT
	4.81 ns
	4.82 ns
	190 ps

	SS
	4.76 ns
	4.77 ns
	240 ps

	FF
	4.86 ns
	4.87 ns
	140 ps

	SF
	4.81 ns
	4.82 ns
	190 ps

	FS
	4.82 ns
	4.83 ns
	180 ps


Figure 12 – Setup Time Summary over all Processes

Hold Time:

To determine the hold time of the register, the input ‘d’ delay is set to the point where it drops at the same time that the clock rises.  If the register failed to latch, the input delay is increased until the register latches the value.  The difference between the time at which the register latches the value and the time that the clock went high is the hold time.  Figure 13 is a hold time simulation for which the hold time is set to zero.  It can be clearly seen that when the hold time is zero, the register still latches its value!  This suggests that the hold time for the SS process is in fact zero.

Figure 13 – Hold Time for SS Process

The following table summarizes the hold times over all process models as simulated in Awaves.

	Process
	Hold Time

	TT
	0 ns

	SS
	0 ns

	FF
	0 ns

	SF
	0 ns

	FS
	0 ns


Figure 14 – Hold Time over all Process Types

The simulation for all processes had a hold time of zero.

Min and Max Clock to Q Time:

Figure 15 – Clock to Q delay with no load at the output
Figure 15 shows the clock to Q delay when there is no load connected to the output.  Figure 17 shows the clock to Q delay when a load is connected.  The values for the minimum and maximum values are summarized in figure 16.

	Load Type
	Min
	Max

	No Load
	190 ps
	300 ps

	20fF Load
	310 ps
	500 ps


Figure 16 – Minimum and Maximum Clock to Q Delay (Summary)

Figure 17 – Clock to Q delay with 20fF load at the output

